


Application of Distributed Loads

Introduction

This tutorial was completed using ANSYS 7.0. The purpose of this tutorial is to explain how to apply distributed loads and use element tables to extract data. Please note that this material was also covered in the 'Bicycle Space Frame' tutorial under 'Basic Tutorials'.

A distributed load of 1000 N/m (1 N/mm) will be applied to a solid steel beam with a rectangular cross section as shown in the figure below. The cross-section of the beam is 10mm x 10mm while the modulus of elasticity of the steel is 200GPa.



Preprocessing: Defining the Problem

1. Open preprocessor menu

`/PREP7`

2. Give example a Title

Utility Menu > File > Change Title ...

`/title, Distributed Loading`

3. Create Keypoints

Preprocessor > Modeling > Create > Keypoints > In Active CS

`K, #, x, y`

We are going to define 2 keypoints (the beam vertices) for this structure as given in the following table:

Keypoint	Coordinates (x,y)
1	(0,0)
2	(1000,0)

4. Define Lines

Preprocessor > Modeling > Create > Lines > Lines > Straight Line

`L, K#, K#`

Create a line between Keypoint 1 and Keypoint 2.

5. Define Element Types

Preprocessor > Element Type > Add/Edit/Delete...

For this problem we will use the BEAM3 element. This element has 3 degrees of freedom (translation along the X and Y axis's, and rotation about the Z axis). With only 3 degrees of freedom, the BEAM3 element can only be used in 2D analysis.

6. Define Real Constants

Preprocessor > Real Constants... > Add...

In the 'Real Constants for BEAM3' window, enter the following geometric properties:

- i. Cross-sectional area AREA: 100

- ii. Area Moment of Inertia IZZ: 833.333
- iii. Total beam height HEIGHT: 10

This defines an element with a solid rectangular cross section 10mm x 10mm.

7. Define Element Material Properties

Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic

In the window that appears, enter the following geometric properties for steel:

- i. Young's modulus EX: 200000
- ii. Poisson's Ratio PRXY: 0.3

8. Define Mesh Size

Preprocessor > Meshing > Size Cntrls > ManualSize > Lines > All Lines...

For this example we will use an element length of 100mm.

9. Mesh the frame


Preprocessor > Meshing > Mesh > Lines > click 'Pick All'

10. Plot Elements

Utility Menu > Plot > Elements

You may also wish to turn on element numbering and turn off keypoint numbering

Utility Menu > PlotCtrls > Numbering ...



Solution Phase: Assigning Loads and Solving

1. Define Analysis Type

Solution > Analysis Type > New Analysis > Static

ANTYPE, 0

2. Apply Constraints


Solution > Define Loads > Apply > Structural > Displacement > On Keypoints

Pin Keypoint 1 (ie UX and UY constrained) and fix Keypoint 2 in the y direction (UY constrained).


3. Apply Loads

We will apply a distributed load, of 1000 N/m or 1 N/mm, over the entire length of the beam.

- o Select **Solution > Define Loads > Apply > Structural > Pressure > On Beams**
- o Click 'Pick All' in the 'Apply F/M' window.
- o As shown in the following figure, enter a value of 1 in the field 'VALI Pressure value at node I' then click 'OK'.



The applied loads and constraints should now appear as shown in the figure below.



Note:

To have the constraints and loads appear each time you select 'Replot' you must change some settings. Select **Utility Menu > PlotCtrls > Symbols....** In the window that appears, select 'Pressures' in the pull down menu of the 'Surface Load Symbols' section.


4. Solve the System

Solution > Solve > Current LS
SOLVE

Postprocessing: Viewing the Results

1. Plot Deformed Shape

General Postproc > Plot Results > Deformed Shape
PLDISP, 2



2. Plot Principle stress distribution


As shown previously, we need to use element tables to obtain principle stresses for line elements.

1. Select **General Postproc > Element Table > Define Table**
2. Click 'Add...'
3. In the window that appears
 - a. enter 'SMAXI' in the 'User Label for Item' section
 - b. In the first window in the 'Results Data Item' section scroll down and select 'By sequence num'
 - c. In the second window of the same section, select 'NMISC, '
 - d. In the third window enter '1' anywhere after the comma
4. click 'Apply'
5. Repeat steps 2 to 4 but change 'SMAXI' to 'SMAJ' in step 3a and change '1' to '3' in step 3d.
6. Click 'OK'. The 'Element Table Data' window should now have two variables in it.
7. Click 'Close' in the 'Element Table Data' window.
8. Select: **General Postproc > Plot Results > Line Elem Res...**
9. Select 'SMAXI' from the 'LabI' pull down menu and 'SMAJ' from the 'LabJ' pull down menu

Note:

- ANSYS can only calculate the stress at a single location on the element. For this example, we decided to extract the stresses from the I and J nodes of each element. These are the nodes that are at the ends of each element.
- For this problem, we wanted the principal stresses for the elements. For the BEAM3 element this is categorized as NMISC, 1 for the 'I' nodes and NMISC, 3 for the 'J' nodes. A list of available codes for each element can be found in the ANSYS help files. (ie. type `help BEAM3` in the ANSYS Input window).

As shown in the plot below, the maximum stress occurs in the middle of the beam with a value of 750 MPa.



Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to '**File > Read input from...**' and select the file.

```

/title, Distributed Loading of a Beam
/PREP7

K,1,0,0           ! Define the keypoints
K,2,1000,0

L,1,2            ! Create the line

ET,1,BEAM3       ! Beam3 element type

R,1,100,833.333,10 ! Real constants - area,I,height

MP,EX,1,200000   ! Young's Modulus
MP,PRXY,1,0.33   ! Poisson's ratio

ESIZE,100        ! Mesh size
LMESH,ALL        ! Mesh line

FINISH
/SOLU

ANTYPE,0         ! Static analysis

DK,1,UX,0,,,UY  ! Pin keypoint 1
DK,2,UY,0       ! Roller on keypoint 2

SFBEAM,ALL,1,PRES,1 ! Apply distributed load

SOLVE
FINISH

/POST1

```

```
PLDISP,2                                ! Plot deformed shape
ETABLE,SMAXI,NMISC, 1                    ! Create data for element table
ETABLE,SMAXJ,NMISC, 3
PLLS,SMAXI,SMAXJ,1,0                     ! Plot ETABLE data
```

Reference

<http://www.mece.ualberta.ca/tutorials/ansys>